

# **Algunas Funciones Avanzadas de SPICE**

**Dr. José Ernesto Rayas-Sánchez**

1

## **Algunas Funciones Avanzadas de SPICE**

---

- Fuentes controladas
- Edición de modelos
- Creación de subcircuitos
- Efectos de la temperatura
- Análisis de sensibilidad
- Análisis de ruido

## Fuentes Controladas Lineales

- Tipos:  $i = gv_c$ ,  $v = ev_c$ ,  $i = fi_c$ ,  $v = hi_c$   
donde  $g, e, f$ , y  $h$  son constantes reales,  $v_c$  y  $i_c$  son las señales controladoras
- Sintaxis  
GXXXXXXXX N+ N- NC+ NC- VALOR  
EXXXXXXXX N+ N- NC+ NC- VALOR  
FXXXXXXXX N+ N- VNOMBRE VALOR  
HXXXXXXXX N+ N- VNOMBRE VALOR  
(VNOMBRE es el nombre de la fuente de voltaje por la cual la corriente controladora fluye)

## Ejemplos de Fuentes Controladas Lineales

$E1 = v_{5-3} = 3v_{11}$   
E1 5 3 11 0 3.0

$G2 = i_{1-0} = 0.5 \times 10^{-3} v_5$   
G2 1 0 5 0 0.5MMHO

$Hfvcc = v_{8-15} = 900i_{Vmedir}$   
Hfvcc 8 15 Vmedir 0.9K

$F1 = i_{7-51} = 10i_{Vsens}$   
F1 7 51 VSENS 10

## Fuentes Controladas Polinomiales

- Tipos:  $i = g(v_c)$ ,  $v = e(v_c)$ ,  $i = f(i_c)$ ,  $v = h(i_c)$   
donde  $g$ ,  $e$ ,  $f$ , y  $h$  son funciones polinomiales de una o más variables, y  $v_c$ ,  $i_c$  son las señales controladoras

- Sintaxis

```
GXXXXXXXX N+ N- <POLY (ND) > NC1+ NC1- ... p0 <p1 ... > <IC=... >
EXXXXXXXX N+ N- <POLY (ND) > NC1+ NC1- ... p0 <p1 ... > <IC=... >
FXXXXXXXX N+ N- <POLY (ND) > VN1 <VN2 ... > p0 <p1 ... > <IC=... >
HXXXXXXXX N+ N- <POLY (ND) > VN1 <VN2 ... > p0 <p1 ... > <IC=... >
```

(VN1, VN2, ... son los nombres de las fuentes de voltaje a través de las cuales pasan las corrientes controladoras, ND es la dimensión del polinomio, IC son las condiciones iniciales)

## Fuentes Controladas Polinomiales (cont.)

Interpretación de los coeficientes del polinomio (p0, p1, ...)

- Si ND = 1

$$y = p_0 + (p_1 * a) + (p_2 * a^{**2}) + (p_3 * a^{**3}) + (p_4 * a^{**4}) + (p_5 * a^{**5}) + \dots$$

- Si ND = 2

$$y = p_0 + (p_1 * a) + (p_2 * b) + (p_3 * a^{**2}) + (p_4 * a * b) + (p_5 * b^{**2}) + (p_6 * a^{**3}) + (p_7 * a^{**2} * b) + (p_8 * a * b^{**2}) + (p_9 * b^{**3}) + \dots$$

- Si ND = 3

$$y = p_0 + (p_1 * a) + (p_2 * b) + (p_3 * c) + (p_4 * a^{**2}) + (p_5 * a * b) + (p_6 * a * c) + (p_7 * b^{**2}) + (p_8 * b * c) + (p_9 * c^{**2}) + (p_{10} * a^{**3}) + (p_{11} * a^{**2} * b) + (p_{12} * a^{**2} * c) + (p_{13} * a * b^{**2}) + (p_{14} * a * b * c) + (p_{15} * a * c^{**2}) + (p_{16} * b^{**3}) + (p_{17} * b^{**2} * c) + (p_{18} * b * c^{**2}) + (p_{19} * c^{**3}) + (p_{20} * a^{**4}) + \dots$$

## Ejemplos de Fuentes Controladas Polinomiales

$$GR = i_{17-3} = 10^{-3}(v_{10-3} + 1.5v_{10-3}^2)$$

```
GR 17 3 10 3 0 1M 1.5M
```

$$EX = v_{18} = v_{13} + v_{15} + v_{17}$$

```
EX 18 0 POLY(3) 13 0 15 0 17 0 0 1 1 1
```

$$F1 = i_{12-10} = 1mA - 1.5mAi_{VCC}$$

```
F1 12 10 VCC 1MA -1.5M
```

$$HXY = v_{13-20} = 500i_{VIN1}i_{VIN2}$$

```
HXY 13 20 POLY(2) VIN1 VIN2 0 0 0 0 500
```

## Fuentes Controladas No-Lineales

- Tipo:  $y = b(v_c, i_c)$   
donde  $y$  puede ser un voltaje o una corriente,  $b$  es una función arbitraria de los valores de C.D. de las señales controladoras  $v_c$  y  $i_c$

- Sintaxis

```
BXXXXXXXX N+ N- <I=EXPR> <V=EXPR>
```

- Funciones y operadores disponibles para la expresión:

abs	asinh	cosh	sin			
acos	atan	exp	sinh	+	-	*
acosh	atanh	ln	sqrt	/	^	unary -
asin	cos	log	tan			

## Ejemplos de Fuentes Controladas No-Lineales

$$i_2 = \cos(v_1) + \sin(v_3)$$

B1 0 2 I=cos (v (1) )+sin (v (3) )

$$v_{2-3} = \ln(\cos(\log(v_{1-2}^2))) + v_3 v_1$$

B2 2 3 V=ln (cos (log (v (1, 2) ^2) ) )+v (3) \*v (1)

$$v_5 = 7e^{\pi i v_d}$$

B3 5 0 V=7\*exp (pi\*i (Vd) )

## Modelos

- Un modelo define el comportamiento eléctrico de una parte o componente
- Dependiendo del tipo de dispositivo, los modelos pueden ser definidos mediante:
  - Conjuntos de parámetros (modelos internos en SPICE)
  - Subcircuitos
- Los modelos se almacenan en archivos de texto llamados bibliotecas
- Las bibliotecas pueden ser globales o locales a un diseño
- Las bibliotecas de modelos se pueden anidar

## Modelos Internos en SPICE

---

- Se definen mediante un conjunto de parámetros
- Pueden editarse mediante el comando de SPICE:  
.MODEL nombre tipo\_modelo [parámetro=valor...]
- Los parámetros no alterados toman su valor por omisión
- Dispositivos modelados internamente:
  - Pasivos
  - Activos

## Dispositivos Pasivos

---

- Con modelo interno editable:
  - R: Resistencias
  - C: Capacitancias
  - O: Líneas de transmisión con pérdidas
- Con modelo interno no-editable:
  - L: Inductancias
  - K: Inductancias mútuas (transformadores)
  - T: Líneas de transmisión sin pérdidas

## Dispositivos Activos

---

- Con modelo interno editable:
  - D: Diodos
  - Q: Transistores bipolares (BJTs)
  - J: Transistores de unión de efecto de campo (JFETs)
  - M: Transistores MOSFET
  - Z: Transistores MESFETS (GaAs FETs)
  - S: Interruptores controlados por voltaje
  - W: Interruptores controlados por corriente


## Tipos de Modelos Internos

---

- RES o R                    Semiconductor resistor model
- CAP o C                    Semiconductor capacitor model
- SWVSWITCH                Voltage controlled switch
- CSWISWITCH                Current controlled switch
- URC                         Uniform distributed RC model
- LTRA                        Lossy transmission line model
- D                             Diode model
- NPN                         NPN BJT model
- PNP                         PNP BJT model
- NJF                         N-channel JFET model
- PJF                         P-channel JFET model
- NMOS                        N-channel MOSFET model
- PMOS                        P-channel MOSFET model
- NMF                         N-channel MESFET model
- PMF                         P-channel MESFET model

## Descripción Detallada de Modelos Internos

Antognetti, Paolo, A. and  
Massobrio, Giuseppe,  
*Semiconductor Device  
Modeling with Spice*.  
McGraw Hill, 1993.



Dr. J. E. Rayas-Sánchez

15

## Ejemplo 1: Modelado de Resistencias

- Forma general en lenguaje SPICE:  
Rnombre Np Nn Modelo valor  
donde  
Rnombre: nombre de la resistencia (por ej. RL).  
NA: nodo de la terminal “positiva”  
NC: nodo de la terminal “negativa”  
Modelo: nombre del modelo  
valor: valor de la resistencia en Ohms
- Ejemplos:  
Rcarga 5 9 ResModeloX 1K  
.model ResModelX RES (R=1 TC1=0.12 TC2=0.05)  
RE e 0 3.3K TC1=0.001 TC2=0.05

Dr. J. E. Rayas-Sánchez

16



## Modelo de las Resistencias

PARÁMETROS DEL MODELO DE RESISTENCIAS (RES)			
Nombre	Parámetro	Unidades	Default
R	Factor multiplicativo	-	1
TC1	Coefficiente lineal de temperatura	1/°C	0
TC2	Coefficiente cuadrático de temperatura	1/°C <sup>2</sup>	0
TCE	Coefficiente exponencial de temperatura	%/°C	0


$$R_{valor} = (\text{valor})R[1 + TC1(T - T_0) + TC2(T - T_0)^2] \quad \text{si TCE} = 0$$

$$R_{valor} = (\text{valor})R[e^{\frac{TCE}{100}(T - T_0)}] \quad \text{si TCE} \neq 0$$


$T$ : temperatura de operación

$T_0$ : temperatura de referencia (*room temperature*)

## Resistencias – Variación con Temperatura



## Resistencias – Variación con Temperatura (cont.)



Dr. J. E. Rayas-Sánchez

19

## Ejemplo 2: Modelado de Diodos

- Forma general en lenguaje SPICE:  
Dname NA NC modname [area] [OFF] [IC=Vd]  
donde
  - Dname: nombre del diodo (por ej. D3).
  - NA: nodo del ánodo
  - NC: nodo del cátodo
  - modname: nombre del modelo
  - area: factor de escala (se asume igual a 1)
  - OFF: para apagarlo como condición inicial en C.D.
  - IC: voltaje inicial del diodo, en el análisis transitorio

Dr. J. E. Rayas-Sánchez

20

## Ejemplo 2: Modelado de Diodos (cont)

- Ejemplos  

```

DIRECT 1 2 DPWR
D4 3 4 DN753 OFF
        
```
- En el caso anterior, D4 utiliza el modelo DN753, que corresponde a un diodo zener 1N753, de 6.2 volts, el cual se define mediante:  


```

.MODEL DN753 D (RS=4.68 BV=6.10 CJO=346P
+ TT=50N M=0.33 VJ=0.75 IS=1E-11 N=1.27
+ IBV=20MA)
        
```

## Modelo del Diodo

PARÁMETROS DEL MODELO DEL DIODO			
Nombre	Parámetro	Unidades	Default
IS	Corriente de saturación	Ampers	1E-14
RS	Resistencia Ohmica	Ohms	0
N	Coefficiente de emisión	-	1
TT	Tiempo de transición	Segundos	0
CJO	Capacitancia de la unión, de polarización cero	Faradios	0
VJ	Potencial de la unión o potencial de contacto	Volts	1
M	Coef. de graduación de la contaminación	-	0.5
EG	Vacío o hueco de energía	e-volts	1.11
XTI	Exponente de la temperatura de la corriente de saturación	-	3
KF	Coefficiente de ruido Flicker	-	0
AF	Exponente de ruido Flicker	-	1
FC	Coefficiente para la capacitancia de agotamiento de polarización directa	-	0.5
BV	Voltaje de rompimiento de reversa	Volts	$\infty$
IBV	Corriente en el voltaje de rompimiento	Ampers	1E-3

## Modelo del Diodo (cont)



- VD Voltaje en el diodo intrínseco únicamente.
- VT Voltaje equivalente de temperatura (igual a  $kT/q$ , 26mV a 27 °C)
- To Temperatura nominal (la cual se establece con la opción TNOM)
- T Temperatura
- q Carga del electrón, en Coulombs.
- k Constante de Boltzman
- Ts Tiempo de almacenamiento de portadores de carga
- f Frecuencia
- df Intervalo de frecuencia
- In Ruido espectral
- $E_G$  Vacío de energía del material, en electrón-volts.

- Ecuaciones estáticas:

$$I_D = I_S \left( e^{\frac{VD}{VT}} - 1 \right) \quad I_R = I_{BV} \cdot e^{\frac{-(VD+BV)}{VT}}$$

Dr. J. E. Rayas-Sánchez

23

## Modelo del Diodo (cont)

- Capacitancias parásitas:

$$CD = \frac{CJO}{(1 - FC)^{(1+M)}} \left[ 1 - FC(1 + M) + M \frac{VD}{VJ} \right]$$

donde

$$C = area \cdot (CT + CD)$$

y

$$CT = TT \frac{dID}{dVD}$$

cuando  $VD > FC \cdot VJ$ :

$$CD = \frac{CJO}{\left( 1 - \frac{VD}{VJ} \right)^M}$$

Dr. J. E. Rayas-Sánchez

24

## Modelo del Diodo (cont)

- Almacenamiento de carga debido a la inyección de portadores minoritarios:

$$QS = TT \cdot IS \left( e^{\frac{VD}{N \cdot VT}} - 1 \right)$$

donde

$$TT = \frac{Ts}{\ln \left( 1 + \frac{ID}{IR} \right)}$$

## Modelo del Diodo (cont)

- Ruido Flicker y ruido térmico,  
Generado por el diodo intrínseco:

$$In^2 = 2q \cdot ID + \frac{kF}{f} (ID)^{AF}$$

Generado por la resistencia parásita Rs:

$$In^2 = \frac{4kT(area)}{RS}$$

- Dependencia de la corriente de saturación con la temperatura

$$I_S(T_1) = I_S(T_0) \left( \frac{T_1}{T_0} \right)^{\frac{XTI}{N}} \exp \left( - \frac{qE_G}{NkT_1} \left( 1 - \frac{T_1}{T_0} \right) \right)$$

## Partes de Fabricantes (Ejemplo)

---

- Transistores bipolares de RF de Motorola
  - MOTOR\_RF.OLB (partes)
  - MOTOR\_RF.LIB (modelos)

## Subcircuitos


---

- Especie de subrutina que el circuito principal “llama”
- Forma de declararlos:  


```
.SUBCKT NombreDelSubcircuito nodo [nodo...]  
topología del circuito  
.ENDS
```
- Forma de llamarlos:  

```
X<NombreDelDispositivo> nodo [nodo...] NombreDelSubcircuito
```

## Ejemplo de Aplicación de Subcircuitos



Equivalente de un Op-Amp casi ideal



```

INVERSOR CON OP-AMP CASI IDEAL
*
VS 1 0 DC 3 AC 1 SIN(0 3 1E3)
RI 1 2 10K
RF 2 3 100K
XA01 0 2 3 OpAmpCasiIdeal
*
* SUBCIRCUITO PARA UN OP-AMP
* CASI IDEAL
.SUBCKT OpAmpCasiIdeal P N OUT
Ri P N 10MEGAHMS
Ro I OUT 75OHMS
Eo I 0 P N 100E3
.ENDS
*
.AC DEC 100 10 10MEG
.TRAN 10us 3ms
.DC VS -1 1 0.1
*
.END
    
```

Dr. J. E. Rayas-Sánchez

29

## Análisis de Sensibilidad

- Se emplea para calcular la sensibilidad de una variable de salida del circuito con respecto a los parámetros de cada uno de los componentes del circuito, en el punto de operación
- Muy útil para determinar tolerancias
- Tres definiciones:

$$\text{Sensibilidad } S_x^y = \frac{\partial y}{\partial x}$$

Sensibilidad normalizada

$$S_x^y = \frac{\partial \ln y}{\partial \ln x} = \frac{x}{y} \frac{\partial y}{\partial x}$$

Sensibilidad seminormalizada

$$S_x^y = \frac{\partial y}{\partial \ln x} = x \frac{\partial y}{\partial x} \quad S_x^y = \frac{\partial \ln y}{\partial x} = \frac{1}{y} \frac{\partial y}{\partial x}$$

Dr. J. E. Rayas-Sánchez

30

## Análisis de Sensibilidad (cont)


- Se implementa mediante el comando de SPICE:  
`.SENS OUTVAR`
- SPICE calcula la sensibilidad (Berkeley) y la sensibilidad seminormalizada en porcentaje (OrCad Cadence):  

$$\frac{\partial y}{\partial x} \quad y \quad \frac{x}{100} \frac{\partial y}{\partial x}$$
- El SPICE de Berkeley soporta análisis de sensibilidad de C.D. y de señal pequeña (C.A.). PSpice de OrCad soporta únicamente análisis de sensibilidad de C.D.
- Ejemplos:  
`.SENS V(3,OUT)`  
`.SENS V(OUTPUT) AC DEC 10 100 100k`  
`.SENS I(VTEST)`

Dr. J. E. Rayas-Sánchez

31

## Análisis de Sensibilidad – Ilustración




Dr. J. E. Rayas-Sánchez

32



## Análisis de Sensibilidad – Ilustración



$$S_{R_1}^{v_o} = ? \quad S_{R_2}^{v_o} = ? \quad S_{V_S}^{v_o} = ?$$

$$v_o = \frac{V_S(R_2)}{(R_1 + R_2)}$$

$$\frac{\partial v_o}{\partial R_1} = \frac{-V_S R_2}{(R_1 + R_2)^2} = \frac{-(3K)}{(4K)^2} = -0.1875 \text{mV}/\Omega$$


$$\frac{\partial v_o}{\partial R_2} = \frac{-V_S(R_2)}{(R_1 + R_2)^2} + \frac{V_S}{(R_1 + R_2)} = \frac{V_S R_1}{(R_1 + R_2)^2} = \frac{1K}{(4K)^2} = +0.0625 \text{mV}/\Omega$$

$$\frac{\partial v_o}{\partial V_S} = \frac{R_2}{R_1 + R_2} = \frac{3K}{4K} = +0.75 \text{V}/\text{V}$$

Dr. J. E. Rayas-Sánchez


33

## Análisis de Sensibilidad – Ilustración (cont)



Example on Sensitivities


```
*-----
Vs   in  0  DC  1
R1   in  out 1K
R2   out 0   3K
.control
SENS v(out)
print all
.endc
.end
```



Dr. J. E. Rayas-Sánchez

34

## Análisis de Sensibilidad – Ilustración (cont)



```
Sensitivity_output.txt - Bloc de notas
Archivo Edición Formato Ver Ayuda
Title: Example on Sensitivities
Date: wed Mar 17 17:56:27 2010
Plotname: Sensitivity Analysis
Flags: real
No. Variables: 3
No. Points: 1
Variables:
0          r2      voltage
1          r1      voltage
2          vs      voltage
values:
0          6.249993749672883e-05
          -1.874998125105153e-04
          7.49999999383001e-01
```


Dr. J. E. Rayas-Sánchez

35

Example on Sensitivities

```
*-----
Vs  in  0  DC  1
R1  in  out 1K
R2  out 0   3K
.control
SENS v(out)
write Sensitivity_output.txt
.endc
.end
```

## Análisis de Sensibilidad – Ilustración (cont)



```
Sensitivity_output.csv - Bloc de notas
Archivo Edición Formato Ver Ayuda
r2,r1,vs
6.249993749672883e-05,-1.874998125105153e-04,7.49999999383001e-01
```

Dr. J. E. Rayas-Sánchez

36

Example on Sensitivities

```
*-----
Vs  in  0  DC  1
R1  in  out 1K
R2  out 0   3K
.control
SENS v(out)
write Sensitivity_output.csv
.endc
.end
```

## Análisis de Ruido




- Dada una señal de entrada (de C.A.) y un puerto de salida, el análisis de ruido de SPICE calcula las contribuciones de ruido de cada dispositivo generador de ruido en el puerto de salida a cada frecuencia simulada
- Los únicos generadores de ruido para SPICE son las resistencias y los dispositivos semiconductores
- Las clases de ruido que SPICE considera son el ruido térmico, el ruido Flicker, y el ruido *shot*

Dr. J. E. Rayas-Sánchez



37

## Tipos de Ruido

- Ruido Térmico: producido por el movimiento aleatorio de los electrones en conductores



- Ruido Flicker: producido por vibraciones en los enlaces atómicos cercanos a la interfase entre semiconductores y dieléctricos



- Ruido Shot: Producido por el paso de corriente directa a través de barreras de potencial (uniones p-n)

Dr. J. E. Rayas-Sánchez

38

## Análisis de Ruido en SPICE

---

- Formas de ejecutarlo:

```
.NOISE V(output) SRC (DEC|LIN|OCT) PTS Fstart Fstop
```

```
.AC (DEC|LIN|OCT) PTS Fstart Fstop
```

```
.NOISE V(output) SRC Interval
```

SRC es la fuente (de voltaje o de corriente) en donde se calcula el ruido de entrada equivalente (INOISE)

Interval es el intervalo de impresión de resultados numéricos en el archivo de salida (contribuciones individuales de todos los generadores de ruido)

## Análisis de Ruido en SPICE (cont.)

---

- Ejemplos:

```
.NOISE V(6) VIN DEC 10 20HZ 20KHZ
```

```
.NOISE V(7,2) V1 OCT 8 1.0 1.0e6
```

```
.AC DEC 20 10Hz 1GHz
```

```
.NOISE v(out) Vs 20
```


## Análisis de Ruido (cont.)

- Para mostrar los resultados del análisis de ruido:  
.PRINT NOISE *output*  
.PLOT NOISE *output*  
donde *output* puede ser:  
ONoise (ruido total en la salida)  
INoise (ruido a la salida referido a la entrada)
- La gráfica de ONoise representa el voltaje de ruido total a la salida, a cada frecuencia (suma RMS de todas las contribuciones de ruido)
- La gráfica de INoise representa el voltaje de ruido equivalente a la entrada, a cada frecuencia, que generaría a ONoise si el circuito no tuviera generadores de ruido

Dr. J. E. Rayas-Sánchez

41

## Análisis de Ruido – Ilustración



### Noise Calculations

```
Vs 1 0 DC 0 AC 1
R1 1 2 1Kohm
R2 2 0 1Kohm

*.control
*AC dec 10 10 1MEG
*plot v(2)
*.endc

.options TEMP = 27
.NOISE v(2) Vs dec 10 10 1MEG
.plot noise onoise inoise


.end
```

Dr. J. E. Rayas-Sánchez

42

## Análisis de Ruido – Ilustración (cont.)

### Resultados del análisis en CA




Dr. J. E. Rayas-Sánchez

43

## Análisis de Ruido – Ilustración (cont.)

### Resultados del análisis de ruido a 27 °C



Dr. J. E. Rayas-Sánchez

44

## Análisis de Ruido – Ilustración (cont.)

Resultados del análisis de ruido a 150 °C

